
Το αυτοκίνητο
Στόχος της δραστηριότητας: Να χρησιμοποιήσουν τη γλώσσα Logo για να κατασκευάσουν γραφικά σχέδια βασιζόμενα σε επαναλαμβανόμενες διαδικασίες.

Επίσης, να μπορούν τα παιδιά να ορίζουν διαδικασίες με μεταβλητές για τη σχεδίαση σχημάτων συγκεκριμένου μεγέθους.

Γνωριμία με τις δυνατότητες του προγράμματος: Στη δραστηριότητα αυτή τα παιδιά χρησιμοποιούν τη γλώσσα Logo για να ορίσουν τις διαδικασίες επαναλαμβανόμενων σχημάτων που λειτουργούν ως κατασκευαστικά μέρη στην σχεδίαση ενός αυτοκινήτου. Έτσι τα παιδιά θα μάθουν ότι μπορούν να κατασκευάσουν ακόμα και σύνθετα γραφικά με την επανάληψη συγκεκριμένων σχεδίων.
Προϋπάρχουσα γνώση γι’ αυτή την δραστηριότητα θεωρείται η γνώση των βασικών εντολών της Logo καθώς επίσης και η διαδικασία χρωματισμού των σχεδίων

Προεργασία και μεθοδολογία: Φτιάξτε μια φωτοτυπία για κάθε ζευγάρι παιδιών που θα δουλέψει στη συνέχεια στον υπολογιστή, η οποία θα απεικονίζει την πλάγια όψη ενός αυτοκινήτου σ’ ένα δρόμο και θα ορίζει τα διάφορα μέρη για τα οποία οι μαθητές θα προσπαθήσουν να φτιάξουν διαδικασίες για τη σχεδίασή τους, δηλαδή μπροστινό μέρος αυτοκινήτου, πίσω μέρος, ρόδα, δρόμος, διαχωριστική λωρίδα.

Ζητήστε τους να γράψουν εντολές με τις οποίες θα κατασκευάσουν τις παραπάνω διαδικασίες, έτσι ώστε να σκεφτούν πώς θα χρησιμοποιήσουν απλές εντολές της γλώσσας Logo για να σχεδιάσουν γωνίες, κύκλους, τόξα και να χρωματίσουν. Υπενθυμίστε στα παιδιά πώς να χρησιμοποιήσουν αυτές τις εντολές, fd, bk, rt, lt, repeat, pu, pd, home, setpc, setfc, fill.

Αφήστε τα να πειραματιστούν στον υπολογιστή μ’ αυτές τις εντολές φτιάχνοντας διάφορα σχήματα και στη συνέχεια δώστε τους κι εσείς κάποια απλά παραδείγματα χρήσης αυτών των εντολών που να σχεδιάζουν γωνίες, κύκλους, τόξα και να χρωματίζουν.
Δώστε τους τα πρώτα βήματα σε κάθε ζητούμενη διαδικασία και καθοδηγείστε τα να φτάσουν στην ολοκλήρωση των διαδικασιών.

Ξεκινήστε απ’ την διαδικασία roda η οποία αποτελείται από δυο ομόκεντρους κύκλους που χρωματίζονται με κίτρινο και μαύρο χρώμα. Αυτή θα είναι:

to roda

pd

repeat 360 [fd 30*6.28/360 lt 1]

lt 90

pu

fd 10

rt 90

pd

repeat 360 [fd 20*6.28/360 lt 1]

lt 90

pu

fd 10

setfc [215 204 72]

fill

bk 15

setfc [0 0 0]

fill

home

end
Έπειτα δημιουργείστε το περίγραμμα του πίσω μέρους του αυτοκινήτου, με την διαδικασία pisomeros, το οποίο αποτελείται από γωνίες και τόξα συγκεκριμένων μοιρών:

to pisomeros

lt 90

fd 60

rt 75

repeat 150 [fd 45*6.28/360 lt 1]

rt 75

fd 30

rt 90

fd 30

rt 45

repeat 45 [fd 120*6.28/360 rt 1]

lt 90

repeat 60 [fd 50*6.28/360 rt 1]

rt 30

fd 60

end
Επαναλάβετε το ίδιο με την διαδικασία mprostinomeros, για το μπροστινό μέρος του αυτοκινήτου:

to mprostinomeros

repeat 45 [fd 180*6.28/360 rt 1]

lt 45

repeat 45 [fd 180*6.28/360 rt 1]

rt 45

fd 8

rt 90

fd 20

rt 75

repeat 150 [fd 45*6.28/360 lt 1]

rt 75

fd 140

end
Πληκτρολογώντας τώρα διαδοχικά στην γραμμή εντολών της Logo τις εντολές pisomeros και mprostinomeros θα δημιουργηθεί το περίγραμμα του αυτοκινήτου.

Καθοδηγείστε τα παιδιά να σχεδιάσουν τις πόρτες, κατασκευάζοντας την αντίστοιχη διαδικασία portes, οι οποίες αποτελούνται από ευθείες και γωνίες, δηλαδή απλή κίνηση και στροφή της χελώνας:

to portes

setpensize [2 2]

rt 90

pu

fd 10

setfc [184 10 36]

fill

bk 10

pd

fd 110

bk 50

lt 86

fd 95

lt 90

fd 30

bk 30

rt 90

bk 95

rt 176

fd 120

rt 90

fd 60

bk 60

lt 90

bk 120

lt 30

pu

fd 10

setfc [43 37 92]

fill

lt 120

fd 20

fill

end
Εισάγετε την ιδέα των μεταβλητών και σχεδιάστε τις λωρίδες του δρόμου, με την διαδικασία με μεταβλητή lorides η οποία αφορά την κίνηση της χελώνας σε μια ευθεία δημιουργώντας διακεκομμένες γραμμές, μήκους όσο ορίζει η μεταβλητή:
to lorides :l

fd 140

lt 90

fd 220

setpc [255 255 255]

setpensize [10 10]

repeat 6 [pd bk :l pu bk 100-:l]

home

end
Εξηγήστε στα παιδιά ότι οι διαδικασίες συνήθως δομούνται από άλλες απλούστερες, αλλά επίσης μπορούν να καλέσουν και να φωλιάσουν άλλες διαδικασίες μέσα τους και στη συνέχεια δημιουργείστε τον δρόμο με την διαδικασία dromos η οποία θα περιέχει την διαδικασία lorides και συνεπώς θα πρέπει να εμφανίζει την μεταβλητή της στο όνομά της όπως επίσης και κάθε άλλη διαδικασία που θα περιέχει την lorides. Ο δρόμος θα αποτελείται από ένα απλό παραλληλόγραμμο:
to dromos :d

setfc [92 82 91]

pu

bk 60

lt 90

pd

fd 360

rt 120

fd 400

rt 60

fd 600

rt 120

fd 400

rt 60

fd 240

pu

home

fill

lorides :d

setpensize [1 1]

setfc [0 0 0]

setpc [0 0 0]

pd

end
Σύμφωνα με την τελευταία ανάλυση, τα παιδιά μπορούν πλέον να κατασκευάσουν την τελική διαδικασία autokinito, η οποία θα περιέχει επίσης στο όνομά της την μεταβλητή απ’ την διαδικασία dromos και θα καλεί με την σειρά της τις διαδικασίες dromos, pisomeros, mprostinomeros, portes και roda ώστε να κατασκευάσει το τελικό σχέδιο:

to autokinito :w

dromos :w

setpensize [4 4]

setpc [116 5 22]

pisomeros

mprostinomeros

portes

home

setpensize [1 1]

setfc [0 0 0]

setpc [0 0 0]

lt 90

fd 75

rt 90

roda

rt 90

fd 215

lt 90

roda

pd

end
Έτσι γράφοντας στην γραμμή εντολών της Logo autokinito 60 ή autokinito 40 ή οποιοδήποτε αριθμό που θα καθορίζει το μήκος των διαχωριστικών λωρίδων στο δρόμο, προκύπτει το τελικό σχέδιο.

Περισσότερες δημιουργικές ιδέες: Τα παιδιά μπορούν να δουλέψουν στη συνέχεια για να δώσουν μεταβλητές και στις υπόλοιπες διαδικασίες ώστε να μπορούν να μεταβάλουν το μέγεθος ολόκληρης της κατασκευής.

Επίσης θα μπορούσαν να δημιουργήσουν νέα μοντέλα αυτοκινήτων με νέα πρωτότυπα και πιο σύνθετα σχήματα.
